
LC_transLine_2Speed
*
*           RIN         1     2    3     4      5     6    7     8     9     10
*   VIN                   ___   ___   ___   ___   ___   ___   ___   ___   ___   ___    ___
*       ___/\  /\  /\____|LC1|_|LC1|_|LC1|_|LC1|_|LC1|_|LC2|_|LC2|_|LC2|_|LC2|_|LC2|__|OUT|
*     _|__   \/  \/  _|_ |___| |___| |___| |___| |___| |___| |___| |___| |___| |___|  |___|
*    /_   \          ___  _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_     |__
*   // \   \    CIN   |   ///   ///   ///   ///   ///   ///   ///   ///   ///   ///     | _|_
*   \   \_//         _|_                                                                | ___ COUT
*    \____/          ///                                                __/\  /\  /\____|  |
*     _|_                                                              _|_  \/  \/ ROUT   _|_
*     ///                                                              ///                ///
*              www.idea2ic.com dsauersanjose@aol.com  4/15/08
*

.OPTIONS  GMIN=1e-18    METHOD=euler   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1    
VIN       VIN    0      PWL( 0 0 2n 0 3.0n 1  18n 1 19.0n 0)   
RIN       VIN    1      50
C1        VIN    0      1.6p
XLC1      1      2      LC1
XLC2      2      3      LC1
XLC3      3      4      LC1 
XLC4      4      5      LC1 
XLC5      5      6      LC1 
XLC6      6      7      LC2
XLC7      7      8      LC2
XLC8      8      9      LC2
XLC9      9      10     LC2
XLC10     10     OUT    LC2
R1        OUT    0      50
C2        OUT    0      -.8p
.tran     0.01n   30n 0 30n

*#1===The_Goal_Is_To_simulate_Two_different_cables=====
.control
run
plot     v(1) v(6) out   title   RC_1GHz_and2GHz_50out
.endc

.SUBCKT LC1  IN    OUT
L1      IN   OUT   8n
C1      OUT  0     3.2p
.ENDS   LC1

*                               .SUBCKT LC1 IN OUT
*          _  _  _  L1
*    ___  / \/ \/ \    ___        ___    _______     ___
*   |IN |_| () () |___|OUT|      |IN |__|       |___|OUT|
*   |___|           | |___|      |___|  |  LC1  |   |___|
*                  _|_                  |_______|
*                  ___ C1                  _|_
*                   |                      ///
*                  _|_
*                  ///    8nH_3.2pF =>50Ohms @ 1Ghz  50*3.2p =160ps
*
*    http://www.idea2ic.com/PlayWithJavascript/L_C_R_F.html
*
*

.SUBCKT LC2  IN    OUT
L1      IN   OUT   4n
C1      OUT  0     1.6p
.ENDS   LC1

*                               .SUBCKT LC2 IN OUT
*          _  _  _  L1
*    ___  / \/ \/ \    ___        ___    _______     ___
*   |IN |_| () () |___|OUT|      |IN |__|       |___|OUT|
*   |___|           | |___|      |___|  |  LC2  |   |___|
*                  _|_                  |_______|
*                  ___ C1                  _|_
*                   |                      ///
*                  _|_
*                  ///    4nH_1.6pF =>50Ohms @ 2Ghz  50*1.6p =80ps
*
.end

=====================END_OF_SPICE============================

*#1===The_Goal_Is_To_simulate_Two_different_cables=====



This example is like connecting two different 50 Ohms cables.
The cable may have two different speeds.
Subciruits LC1 and LC2 resonate at two different rates. 
Their LC resonance frequencies differ by a factor of two. 
Both LC1 and LC2 resonate at 50 Ohms

Capacitors C1 and C2 attempt to provide 
symmetrical capacitance to all inductors.

There appears to be a slight reflection at the cable junctions.
The cable speeds still matches the RC time constants.


